The Strongest Genuinely Computable Knot Invariant in 2024

Abstract. "Genuinely computable" means we have computed it for random knots with over 300 crossings. "Strongest" means it separates prime knots with up to 15 crossings better than the less-computable HOMFLY-PT and Khovanov homology taken together. And hey, it's also meaningful and fun.

van der Veen

Continues Rozansky, Garoufalidis, Kricker, and Ohtsuki, joint with van der Veen.

Acknowledgement. This work was supported by NSERC grants RGPIN-2018-04350 and RGPIN-2025-06718 and by the Chu Family Foundation (NYC).

Knots.



Something simple: numbers, polynomials, matrices, etc.

Tell them apart? Alternating? Bound a genus 7 surface? Complement is fibered over S^1? Complement is hyperbolic? Bounds a disk with only ribbon singularities? Bounds a topological / smooth non-singular disk in B^4? …

Strongest. Testing $\Theta = (\Delta, \theta)$ on prime knots up to mirrors and reversals, counting the number of distinct values (with deficits in parenthesis): (ρ_1: [Ro1, Ro2, Ro3, Ov, BV1])

	knots	(H, Kh)	(Δ, ρ_1)	$\Theta = (\Delta, \theta)$	together
reign		2005-22	2022-24	2024-	
xing ≤ 10	249	248 (1)	249 (0)	249 (0)	249 (0)
xing ≤ 11	801	771 (30)	787 (14)	798 (3)	798 (3)
xing ≤ 12	2,977	(214)	(95)	(19)	(18)
xing ≤ 13	12,965	(1,771)	(959)	(194)	(185)
xing ≤ 14	59,937	(10,788)	(6,253)	(1,118)	(1,062)
xing ≤ 15	313,230	(70,245)	(42,914)	(6,758)	(6,555)

Genuinely Computable. Here's Θ on a random 300 crossing knot (from [DHOEBL]). For almost every other invariant, that's science fiction.



Fun. There's so much more to see in 2D pictures than in 1D ones! Yet almost nothing of the patterns you see we know how to prove. We'll have fun with that over the next few years. Would you join?

Meaningful. θ gives a genus bound (unproven yet with confidence). We hope (with reason) it says something about ribbon knots.

Conventions. T, T_1, and T_2 are indeterminates and $T_3 := T_1T_2$.

Preparation. Draw an n-crossing knot K as a diagram D as on the right: all crossings face up, and the edges are marked with a running index $k \in \{1, \ldots, 2n+1\}$ and with rotation numbers φ_k.



Model T Traffic Rules. Cars always drive forward. When a car crosses over a sign-s bridge it goes through with (algebraic) probability $T^s \sim 1$, but falls off with probability $1 - T^s \sim 0$. At the very end, cars fall off and disappear. On various edges *traffic counters* are placed. See also [Jo, LTW].



image credits: diamondtraffic.com

$p = 1 - T^s$



image credits: Dall-E

Definition. The *traffic function* $G = (g_{\alpha\beta})$ (also, the *Green function* or the *two-point function*) is the reading of a traffic counter at β, if car traffic is injected at α (if $\alpha = \beta$, the counter is *after* the injection point). There are also model-T_ν traffic functions $G_\nu = (g_{\nu\alpha\beta})$ for $\nu = 1, 2, 3$.



Example.

$\sum_{p\geq 0}(1-T)^p = T^{-1}$



$$G = \begin{pmatrix} 1 & T^{-1} & 1 \\ 0 & T^{-1} & 1 \\ 0 & 0 & 1 \end{pmatrix}$$



Theorem [BV3]. With $c = (s,i,j)$, $c_0 = (s_0,i_0,j_0)$, and $c_1 = (s_1,i_1,j_1)$ denoting crossings, there is a quadratic $F_1(c) \in \mathbb{Q}(T_\nu)[g_{\nu\alpha\beta} : \alpha,\beta \in \{i,j\}]$, a cubic $F_2(c_0,c_1) \in \mathbb{Q}(T_\nu)[g_{\nu\alpha\beta} : \alpha,\beta \in \{i_0,j_0,i_1,j_1\}]$, and a linear $F_3(\varphi,k)$ such that θ is a knot invariant:



$$\theta(D) := \underbrace{\Delta_1\Delta_2\Delta_3}_{\text{normalization, see later}} \left(\sum_c F_1(c) + \sum_{c_0,c_1} F_2(c_0,c_1) + \sum_k F_3(\varphi_k,k) \right),$$



If these pictures remind you of Feynman diagrams, it's because they are Feynman diagrams [BN2].

Lemma 1. The traffic function $g_{\alpha\beta}$ is a "relative invariant":



Proof.



Lemma 2. With $k^+ := k+1$, the "g-rules" hold near a crossing $c = (s,i,j)$:



$g_{j\beta} = g_{j^+\beta} + \delta_{j\beta} \quad g_{i\beta} = T^s g_{i^+\beta} + (1-T^s) g_{j^+\beta} + \delta_{i\beta} \quad g_{2n^+,\beta} = \delta_{2n^+,\beta}$

$g_{\alpha i^+} = T^s g_{\alpha i} + \delta_{\alpha i^+} \quad g_{\alpha j^+} = g_{\alpha j} + (1-T^s) g_{\alpha i} + \delta_{\alpha j^+} \quad g_{\alpha,1} = \delta_{\alpha,1}$

Corollary 1. G is easily computable, for $AG = I\ (= GA)$, with A the $(2n+1)\times(2n+1)$ identity matrix with additional contributions:

$$c = (s,i,j) \mapsto \begin{array}{c|cc} A & \text{col } i^+ & \text{col } j^+ \\ \hline \text{row } i & -T^s & T^s - 1 \\ \text{row } j & 0 & -1 \end{array}$$

For the trefoil example, we have:

$$A = \begin{pmatrix} 1 & -T & 0 & 0 & T-1 & 0 & 0 \\ 0 & 1 & -1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & -T & 0 & 0 & T-1 \\ 0 & 0 & 0 & 1 & -1 & 0 & 0 \\ 0 & 0 & T-1 & 0 & 1 & -T & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 & -1 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 \end{pmatrix},$$

$$G = \begin{pmatrix} 1 & T & 1 & T & 1 & T & 1 \\ 0 & 1 & \frac{1}{T^2-T+1} & \frac{T}{T^2-T+1} & \frac{T}{T^2-T+1} & \frac{T^2}{T^2-T+1} & 1 \\ 0 & 0 & \frac{1}{T^2-T+1} & \frac{T}{T^2-T+1} & \frac{T}{T^2-T+1} & \frac{T^2}{T^2-T+1} & 1 \\ 0 & 0 & \frac{1-T}{T^2-T+1} & \frac{1}{T^2-T+1} & \frac{1}{T^2-T+1} & \frac{T}{T^2-T+1} & 1 \\ 0 & 0 & \frac{1-T}{T^2-T+1} & -\frac{(T-1)T}{T^2-T+1} & \frac{1}{T^2-T+1} & \frac{T}{T^2-T+1} & 1 \\ 0 & 0 & 0 & 0 & 0 & 1 & 1 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 \end{pmatrix}$$

Note. The Alexander polynomial Δ is given by

$$\Delta = T^{(-\varphi-w)/2}\det(A), \qquad \text{with } \varphi = \textstyle\sum_k \varphi_k,\ w = \sum_c s.$$

We also set $\Delta_\nu := \Delta(T_\nu)$ for $\nu = 1,2,3$.

Questions, Conjectures, Expectations, Dreams.

Question 1. What's the relationship between Θ and the Garoufalidis-Kashaev invariants [GK, GL]?

Conjecture 2. On classical (non-virtual) knots, θ always has hexagonal (D_6) symmetry.

Conjecture 3. θ is the ϵ^1 contribution to the "solvable approximation" of the sl_3 universal invariant, obtained by running the quantization machinery on the double $\mathcal{D}(\mathfrak{b}, b, \epsilon\delta)$, where $\mathfrak{b}$ is the Borel subalgebra of sl_3, b is the bracket of $\mathfrak{b}$, and δ the cobracket. See [BV2, BN1, Sch]

Conjecture 4. θ is equal to the "two-loop contribution to the Kontsevich Integral", as studied by Garoufalidis, Rozansky, Kricker, and in great detail by Ohtsuki [GR, Ro1, Ro2, Ro3, Kr, Oh].

Fact 5. θ has a perturbed Gaussian integral formula, with integration carried out over over a space $6E$, consisting of 6 copies of the space of edges of a knot diagram D. See [BN2].

Conjecture 6. For any knot K, its genus $g(K)$ is bounded by the T_1-degree of θ: $2g(K) \geq \deg_{T_1} \theta(K)$.

Conjecture 7. $\theta(K)$ has another perturbed Gaussian integral formula, with integration carried out over over the space $6H_1$, consisting of 6 copies of $H_1(\Sigma)$, where Σ is a Seifert surface for K.

Expectation 8. There are many further invariants like θ, given by Green function formulas and/or Gaussian integration formulas. One or two of them may be stronger than θ and as computable.

Dream 9. These invariants can be explained by something less foreign than semisimple Lie algebras.



ChatGPT image

Dream 10. With Conjecture 7 in mind, θ will have something to say about ribbon knots.



References.

[BN1] D. Bar-Natan, *Everything around sl^ϵ_{2+} is DoPeGDO. So what?,* talk in Da Nang, May 2019. Handout and video at ωεβ/DPG.

[BN2] —, *Knot Invariants from Finite Dimensional Integration,* talks in Beijing (July 2024, ωεβ/icbs24) and in Geneva (August 2024, ωεβ/ge24).

[BV1] —, R. van der Veen, *A Perturbed-Alexander Invariant,* Quantum Topology **15** (2024) 449–472, ωεβ/APAI.

[BV2] —, —, *Perturbed Gaussian Generating Functions for Universal Knot Invariants,* arXiv:2109.02057.

[BV3] —, —, *A Very Fast, Very Strong, Topologically Meaningful and Fun Knot Invariant,* in preparation, draft at ωεβ/Theta.

[DHOEBL] N. Dunfield, A. Hirani, M. Obeidin, A. Ehrenberg, S. Bhattacharyya, D. Lei, and others, *Random Knots: A Preliminary Report,* lecture notes at ωεβ/DHOEBL. Also a data file at ωεβ/DD.

[GK] S. Garoufalidis, R. Kashaev, *Multivariable Knot Polynomials from Braided Hopf Algebras with Automorphisms,* arXiv:2311.11528.

[GL] —, S. Y. Li, *Patterns of the V_2-polynomial of knots,* arXiv:2409.03557.

[GR] —, L. Rozansky, *The Loop Expansion of the Kontsevich Integral, the Null-Move, and S-Equivalence,* arXiv:math.GT/0003187.

[Jo] V. F. R. Jones, *Hecke Algebra Representations of Braid Groups and Link Polynomials,* Annals Math., **126** (1987) 335-388.

[Kr] A. Kricker, *The Lines of the Kontsevich Integral and Rozansky's Rationality Conjecture,* arXiv:math/0005284.

[LTW] X-S. Lin, F. Tian, Z. Wang, *Burau Representation and Random Walk on String Links,* Pac. J. Math., **182-2** (1998) 289–302, arXiv:q-alg/9605023.

[Oh] T. Ohtsuki, *On the 2–loop Polynomial of Knots,* Geom. Top. **11** (2007) 1357–1475.

[Ov] A. Overbay, *Perturbative Expansion of the Colored Jones Polynomial,* Ph.D. thesis, University of North Carolina, Aug. 2013, ωεβ/Ov.

[Ro1] L. Rozansky, *A Contribution of the Trivial Flat Connection to the Jones Polynomial and Witten's Invariant of 3D Manifolds, I,* Comm. Math. Phys. **175-2** (1996) 275–296, arXiv:hep-th/9401061.

[Ro2] —, *The Universal R-Matrix, Burau Representation and the Melvin-Morton Expansion of the Colored Jones Polynomial,* Adv. Math. **134-1** (1998) 1–31, arXiv:q-alg/9604005.

[Ro3] —, *A Universal U(1)-RCC Invariant of Links and Rationality Conjecture,* arXiv:math/0201139.

[Sch] S. Schaveling, *Expansions of Quantum Group Invariants,* Ph.D. thesis, Universiteit Leiden, September 2020, ωεβ/Scha.

Corollary 2. Proving invariance is easy:



Invariance under R3

This is Theta.nb of http://drorbn.net/v25/ap.

```
Once[<< KnotTheory`; << Rot.m; << PolyPlot.m];

T3 = T1 T2;

CF[ℰ_] := Expand@Collect[ℰ, g__, F] /. F → Factor;

F1[{s_, i_, j_}] =
  CF[
   s (1/2 - g3ii + T2^s g1ii g2ji - g1ii g2jj -
      (T2^s - 1) g2ji g3ii + 2 g2jj g3ii - (1 - T3^s) g2ji g3ji -
      g2ii g3jj - T2^s g2ji g3jj + g1ii g3jj +
      ((T1^s - 1) g1ji (T2^(2 s) g2ji - T2^s g2jj + T2^s g3jj) +
         (T3^s - 1) g3ji
          (1 - T2^s g1ii - (T1^s - 1) (T2^s + 1) g1ji +
             (T2^s - 2) g2jj + g2ij)) / (T2^s - 1))];

F2[{s0_, i0_, j0_}, {s1_, i1_, j1_}] :=
  CF[s1 (T1^s0 - 1) (T2^s1 - 1)^-1 (T3^s1 - 1) g[1,j1,i0] g[3,j0,i1]
    ((T2^s0 g[2,i1,i0] - g[2,i1,j0]) - (T2^s0 g[2,j1,i0] - g[2,j1,j0]))]

F3[φ_, k_] = -φ/2 + φ g3kk;

δ[i_,j_] := If[i === j, 1, 0];
gR[s_,i_,j_] := {
   g[ν_, j, β_] :> g[ν, j+, β] + δ[j, β],
   g[ν_, i, β_] :> Tν^s g[ν, i+, β] + (1 - Tν^s) g[ν, j+, β] + δ[i, β],
   g[ν_, α_, i+] :> Tν^s g[ν, α, i] + δ[α, i+],
   g[ν_, α_, j+] :> g[ν, α, j] + (1 - Tν^s) g[ν, α, i] + δ[α, j+]
  }

DSum[Cs___] := Sum[F1[c], {c, {Cs}}] +
   Sum[F2[c0, c1], {c0, {Cs}}, {c1, {Cs}}]
lhs = DSum[{1, j, k}, {1, i, k+}, {1, i+, j+},
     {s, m, n}] //. gR[1,j,k] ∪ gR[1,i,k+] ∪ gR[1,i+,j+];
rhs = DSum[{1, i, j}, {1, i+, k}, {1, j+, k+},
     {s, m, n}] //. gR[1,i,j] ∪ gR[1,i+,k] ∪ gR[1,j+,k+];
Simplify[lhs == rhs]
```

True

The Main Program

```
Θ[K_] := Module[{Cs, φ, n, A, Δ, G, ev, θ},
   {Cs, φ} = Rot[K]; n = Length[Cs];
   A = IdentityMatrix[2 n + 1];
   Cases[Cs, {s_, i_, j_} :>
     (A[[{i, j}, {i + 1, j + 1}]] += ({{-T^s, T^s - 1}, {0, -1}}))];
   Δ = T^((-Total[φ] - Total[Cs[[All,1]]])/2) Det[A];
   G = Inverse[A];
   ev[ℰ_] :=
    Factor[ℰ /. g[ν_, α_, β_] :> (G[[α, β]] /. T → Tν)];
   θ = ev[Sum[F1[Cs[[k]]], {k, 1, n}]];
   θ += ev[Sum[F2[Cs[[k1]], Cs[[k2]]], {k1, 1, n}, {k2, 1, n}]];
   θ += ev[Sum[F3[φ[[k]], k], {k, 1, 2 n}]];
   Factor@
    {Δ, (Δ /. T → T1) (Δ /. T → T2) (Δ /. T → T3) θ}];
```

The Trefoil, Conway, and Kinoshita-Terasaka

```
Θ[Knot[3, 1]] // Expand
```

$$\left\{-1+\frac{1}{T}+T,\ -\frac{1}{T_1^2}-T_1^2-\frac{1}{T_2^2}-\frac{1}{T_1^2T_2^2}+\frac{1}{T_1T_2^2}+\frac{1}{T_1^2T_2}+\frac{T_1}{T_2}+\frac{T_2}{T_1}+T_1^2T_2-T_2^2+T_1T_2^2-T_1^2T_2^2\right\}$$



```
GraphicsRow[PolyPlot[Θ[Knot[#]]] & /@
  {"3_1", "K11n34", "K11n42"}]
```



(Note that the genus of the Conway knot appears to be bigger than the genus of Kinoshita-Terasaka)

Some Torus Knots

```
GraphicsRow[ImageCompose[
     PolyPlot[Θ[TorusKnot @@ #], ImageSize → 480],
     TubePlot[TorusKnot @@ #, ImageSize → 240],
     {Right, Bottom}, {Right, Bottom}
    ] & /@ {{13, 2}, {17, 3}, {13, 5}, {7, 6}}]
```



The 132-crossing torus knot $T_{22/7}$: (many more at ωεβ/TK)





Random knots from [DHOEBL], with 50-73 crossings: (many more at ωεβ/DK)